
Массивы 
Описание, вывод, заполнение 
Вручную С помощью стандартных методов 
var a: array of integer; 
a := new integer[5]; 
a[0]:=2; a[1]:=4; a[2]:=6; 
a[3]:=1; a[4]:=3; 
for var i:=0 to a.Length-1 do 
  Print(a[i]); 
foreach var x in a do 
  Println(x); 

var a := Arr(2,4,6,1,3); 
 
 
 
a.Print; 
 
a.PrintLines; 

 
Заполнение случайными 
Вручную С помощью стандартных методов 
var a := new integer[n]; 
for var i:=0 to n-1 do 
  a[i] := Random(1,9); 

var a := ArrRandomInteger(n,1,9); 
 

 
Заполнение по заданному правилу 
Вручную С помощью стандартных методов 
var a := new integer[n]; 
for var i:=0 to n-1 do 
  a[i] := i * i; 

var a := ArrGen(n,i -> i * i); 
 

 
Заполнение арифметической прогрессией 
Вручную С помощью стандартных методов 
var a := new integer[n]; 
a[0] := a0; 
for var i:=1 to n-1 do 
  a[i] := a[i-1] + h; 

var a := ArrGen(n,a0,x->x+h); 
 

 
Заполнение числами Фибоначчи 
Вручную С помощью стандартных методов 
var a := new integer[n]; 
a[0] := 1; a[1] := 1;  
for var i:=2 to n-1 do 
  a[i] := a[i-1] + a[i-2]; 

var a := ArrGen(n,1,1,(x,y)->x+y); 
 

 
Сумма, произведение, среднее 
Вручную С помощью стандартных методов 
var n := a.Length; 
var s := 0; 
for var i:=0 to n-1 do 
  s += a[i]; 
var avg := s/n; 
var p := 1; 
for var i:=0 to n-1 do 
  p *= a[i]; 

var s := a.Sum; 
 
 
 
var avg := a.Average; 
Нет простого аналога 

 
  



Минимум, индекс минимального 
Вручную С помощью стандартных методов 
var min := a[0];  
var ind := 0; 
for var i:=1 to a.Length-1 do 
  if a[i]<min then 
  begin 
    min := a[i]; 
    ind := i; 
  end; 

var min := a.Min; 
var ind := a.IndexMin; 
 

 
Поиск элемента 
Вручную С помощью стандартных методов 
var found := False;  
for var i:=0 to a.Length-1 do 
  if a[i]=x then 
  begin 
    found := True; 
    break; 
  end; 

var found := x in a; 
 
 

 
Поиск индекса 
Вручную С помощью стандартных методов 
var ind := -1;  
for var i:=0 to a.Length-1 do 
  if a[i]=x then 
  begin 
    ind := i; 
    break; 
  end; 

var ind := a.IndexOf(x); 
 
 

var ind := -1;  
for var i:= a.Length-1 downto 0 do 
  if a[i]=x then 
  begin 
    ind := i; 
    break; 
  end; 

var ind := a.LastIndexOf(x); 
 
 

var ind := -1;  
for var i:= a.Length-1 downto 0 do 
  if a[i]>5 then 
  begin 
    ind := i; 
    break; 
  end; 

var ind := a.FindIndex(x->x>5); 
 
 

 
Инвертирование массива 
Вручную С помощью стандартных методов 
var n := a.Length; 
for var i:=0 to n div 2 - 1 do 
  Swap(a[i],a[n-i-1]); 

Reverse(a); 
  или 
a := a[::-1]; 

 
  



Сортировка массива 
Вручную С помощью стандартных методов 
var n := a.Length; 
for var i := 0 to n - 2 do 
for var j := n - 1 downto i + 1 do 
  if a[j] < a[j - 1] then 
    Swap(a[j], a[j - 1]); 

Sort(a); // и она быстрее ! 
 
 

 
Поиск всех по условию 
Вручную С помощью стандартных методов 
var n := a.Length; 
var b := new integer[n]; 
var ib := 0; 
for var i := 0 to n - 1 do 
  if a[i] mod 2 = 0 then 
  begin 
    b[ib] := a[i]; 
    ib += 1; 
  end; 
SetLength(b,ib); 

var b := a.FindAll(x->x mod 2 = 0);  
 

 
Замена 
Вручную С помощью стандартных методов 
for var i := 0 to a.Length - 1 do 
  if a[i] = 2 then 
    a[i] := 3; 

a.Replace(2,3);  
 

 
Преобразование 
Вручную С помощью стандартных методов 
for var i := 0 to a.Length - 1 do 
  a[i] = a[i] * 2; 

a.Transform(x->x*2); 
 

 
Циклический сдвиг влево 
Вручную С помощью стандартных методов 
for var i := 0 to a.Length - 2 do 
  a[i] := a[i+1]; 

a := a[1:] + Arr(a[0]); 
 

 
Циклический сдвиг влево на k 
Вручную С помощью стандартных методов 
var n := a.Length; 
var b := new integer[n]; 
for var i:=k to n-1 do 
  b[i-k] := a[i]; 
for var i:=0 to k-1 do 
  b[i+n-k] := a[i]; 
a := b; 

a := a[k:] + a[:k]; 
 

 
Количество элементов, удовлетворяющих условию 
Вручную С помощью стандартных методов 
var cnt := 0; 
for var i := 0 to a.Length - 1 do 
  if a[i] mod 2 = 0 then 
    cnt += 1; 

var cnt := a.Count(x->x mod 2 = 0); 
 

 


